

SiC パワーデバイス

熱モデルの使い方

SPICE モデルには、熱に関するシミュレーションを行うための熱モデルと呼ばれるものがあります。熱モデルを用いたシミュレーションは、熱設計の初期段階に大まかな見積もりを立てるために実施されます。このアプリケーションノートでは熱モデルの使い方について説明しています。

熱モデルの入手

熱モデルはロームのホームページからダウンロード可能です。サイト検索窓で形名を検索し、検索結果から該当形名をクリックします。個別製品ページに移動しますので、「モデルとツール」タブのデザインモデルから熱計算モデルをダウンロードしてください。

熱モデルの概要

熱モデルの一例を Figure 1 に示します。これは 1200V Nch SiC MOSFET SCT3040KR のもので、同様の図がデータシートにも記載されています。Tj ピンはジャンクション温度、Tc ピンはケース温度、Ta ピンは周囲環境温度を表しています。




Figure 1. 熱モデルの一例、Cauer 型 RC 熱回路網
ROHM 製 SiC MOSFET: SCT3040KR

また、ダウンロードした熱モデルのネットリストを Figure 3 に示します。このネットリストには、SiC MOSFET の電気モデルと熱モデルの両方が記載されています。回路図で示すと Figure 2 のような素子とピンで構成されています。

さらにこのモデルでは SiC MOSFET の電気モデルで発生した消費電力（自己発熱）を熱モデルへ伝達する機能を含んでいます。このため、普段どおりに電気回路のシミュレーションを実行すると Tj ピンにジャンクション温度に相当する電圧が発生します。

次に簡単にネットリストの内容を見て行きます。SiC MOSFET で発生した消費電力を熱モデルへ伝達する機能を説明するために、ネットリストの主要な部分だけを抜き出した回路を Figure 4 に示します。3 つ

のブロックに分かれていますが、ブロック A は SiC MOSFET 部分、ブロック B はボディーダイオード部分、ブロック C は熱抵抗のモデルになります。

Figure 3 のネットリストと見比べると、行 13 から 42 がブロック A、行 44 から 53 がブロック B、行 55 から 62 がブロック C に対応しています。行 55 の E20（電圧制御電圧源）と行 56 の R20（抵抗）は、SiC MOSFET とボディーダイオードで発生した消費電力を熱抵抗モデルの Tj ピンへ電流量として伝達するためのコンポーネントです。通常 Tj ピンへは電流源を使って消費電力に相当する電流値を供給しますが、電流源は出力インピーダンスが高く、初期値の収束性が悪くなることもあるため、電圧源と出力抵抗を組み合わせで低出力インピーダンスの電流源を表現しています。出力インピーダンス R20 の 1kΩ は熱抵抗に対して十分に大きな値です。




Figure 2. 熱計算モデルに含まれている素子とピン構成
ROHM 製 SiC MOSFET: SCT3040KR

次に SiC MOSFET とボディーダイオードで発生する消費電力を検出する部分を説明します。これはネットリスト行 55 の E20 内に記述している計算式で行っています。この部分を式(1)に抜き出します。

$$P_D = (I(V1) - I(V11)) \times V(1,3) + I(V2) \times V(2,3) \quad (1)$$

```

1 * SCT3040KR_T
2 * SiC NMOSFET with driver source Self-heating Thermal model
3 * 1200V 55A 40mOhm
4 * Model Generated by ROHM
5 * All Rights Reserved
6 * Commercial Use or Resale Restricted
7 * Date: 2019/07/09
8 *****D G S DS Tj Tc Ta
9 .SUBCKT SCT3040KR_T 1 2 3 4 Tj Tc Ta
10 .PARAM T0=25 T1=-100 T2=300
11 .FUNC K1(T) {MIN(MAX(T, T1), T2)}
12 *****
13 .FUNC R1(I, T) {30.33m*I*EXP((T-T0)/166.5)+4.822u*I*ABS(I)**1.596*EXP((T-T0)/892.6)}
14 .FUNC R2(I) {I*(1.3*(1+TANH(I))/2+3.7*(1-TANH(I))/2)}
15 .FUNC V1(I, T) {139.4m*ASINH(I/0.1277)*EXP((T-T0)/-409.0)+
16 + 559.5m*ASINH(I/3.997)*EXP((T-T0)/-1244)+25.41m*I*EXP((T-T0)/498.5)}
17 .FUNC V2(V, T) {21.45p*V**13.56*EXP((T-T0)/38.52*EXP((T-T0)/-1793))}
18 .FUNC V3(V, T) {94.67*SINH(V/94.67)*EXP((T-T0)/11811)}
19 .FUNC I1(V, W, T) {V3(V, T)*(1+3.258*(TANH((V3(V, T)-66.90)/113.2)+1)*EXP((T-T0)/-174.1)/10)*
20 + W/(ABS(W)+3.258*(TANH((V3(V, T)-66.90)/113.2)+1)*EXP((T-T0)/-174.1))}
21 .FUNC C1(U, V, W) {(559.5*V+1446*(1-576.3*TANH(W/576.3)/1.345)**-0.5204)*(1+(TANH((U+8.327)/3.706)-1)/5.183)}
22 V1 1 11 0
23 E1 11 12 VALUE={R1(I(V1), K1(V(Tj)))}
24 V2 2 21 0
25 E2 21 22 VALUE={R2(I(V2))}
26 L1 3 32 7.0n
27 R2 3 32 10
28 R5 4 32 1m
29 E6 41 0 VALUE={MIN(MAX(V(22, 32), 0), 22)}
30 V6 41 42 0
31 E7 42 43 VALUE={V1(I(V6), K1(V(Tj)))}
32 G6 43 0 VALUE={MIN(MAX(V2(MAX(V(43), 0), K1(V(Tj))), -500), 500)}
33 C6 43 0 1p
34 R6 43 0 1T
35 G1 12 32 VALUE={MIN(MAX(I1(MAX(I(V6), 0), V(12, 32), K1(V(Tj))), -300), 300)}
36 C7 12 32 1p
37 R3 12 32 1T
38 V3 22 23 0
39 C1 23 12 1p
40 G2 22 12 VALUE={I(V3)*C1(V(22, 12), MIN(MAX(V(22, 12), 0), 0.525), MIN(V(22, 12), 0))}
41 C2 22 32 1.531n
42 R4 22 3 1G
43 *****
44 .FUNC R11(I, T) {87.51m*ASINH(I/0.4297)*EXP((T-T0)/304.7)+43.40m*I*EXP((T-T0)/303.9)}
45 .FUNC I11(V, T) {17.41n*(EXP(V/0.09161/EXP((T-T0)/-311.1))-1)*EXP((T-T0)/-64.07*EXP((T-T0)/761.8))-
46 + 1u*TANH(-V/0.1)*EXP((T-T0)/180.3)-8.756E-30*(EXP(-V/25)-1)*EXP((T-T0)/-200)}
47 .FUNC C11(V, W) {832.8*(V-1.091)+1065*(1-596.7*TANH(W/596.7)/2.183)**-0.5737}
48 V11 32 51 0
49 E11 51 52 VALUE={R11(I(V11), K1(V(Tj)))}
50 V12 52 53 0
51 C11 53 1 1p
52 G11 52 1 VALUE={MIN(MAX(I11(V(52, 1), K1(V(Tj))), -300), 300)+I(V12)*C11(MAX(V(52, 1), 1.091), MIN(V(52, 1), 1.091))}
53 R11 52 1 1T
54 *****
55 E20 T0 Ta VALUE={1k*MIN(MAX(IF(TIME>0, (I(V1)-I(V11))*V(1, 3)+I(V2)*V(2, 3), 0), -2MEG), 2MEG)}
56 R20 T0 Tj 1k
57 C21 Tj Ta 1.385m
58 C22 T2 Ta 10.02m
59 C23 T3 Ta 35.72m
60 R21 Tj T2 25.59m
61 R22 T2 T3 194.8m
62 R23 T3 Tc 219.6m
63 .ENDS SCT3040KR_T

```

Figure 3. 熱モデルネットリストの一例
ROHM 製 SiC MOSFET: SCT3040KR





Figure 4. SiC MOSFET で発生した消費電力を熱モデルへ伝達する機能を説明するために、ネットリストの主要な部分だけを抜き出した回路


ROHM 製 SiC MOSFET: SCT3040KR

第1項で $I(V1)$ は SiC MOSFET のドレイン・ソース間に流れる電流、 $I(V11)$ はボディーダイオードに流れる電流です。それぞれを加算した電流値 ($V11$ の極性に注意) にドレイン・ソース間電圧 $V(1,3)$ を乗じてドレイン・ソース間の消費電力を計算しています。第2項は、 $I(V2)$ がゲート・ソース間に流れる電流で、 $V(2,3)$ はゲート・ソース間電圧です。両者に乗じることでゲート・ソース間の損失を計算しています。第1項と第2項の数値を加算することで全体の消費電力を検出しています。



シミュレーション回路の概要

熱モデルを用いた一般的なシミュレーション回路を Figure 5 に示します。そのときのデバイス構成を Figure 6 に示します。周囲環境温度 T_A をバイアス電圧 V_{BIAS} として T_A ピンに与え、デバイスの消費電力 P_D を電流 I として T_j ピンに印加することで、RC 時定数を持った電圧が T_j ピンに発生します。この電圧がジャンクション温度です。また T_c ピンに接続されている抵抗は、 R_1 がケースとヒートシンク間の熱抵抗 R_{thCF} 、 R_2 がヒートシンクと周囲環境温度間の熱抵抗 R_{thFA} です。なお R_{thCF} は熱界面材 (TIM: Thermal Interface Material) の熱抵抗と接触熱抵抗を含みます。また C_1 はヒートシンクの熱容量で、 R_2 と C_1 でヒートシンクを構成しています。



シミュレーション方法

使用するシミュレータによって設定方法は異なりますが、ここでは SystemVision®Cloud (Mentor Graphics Corp.) を使用して説明します。

まずネットリストを参照して回路シンボルを作成します。ネットリストの行 8 から 9 にピン情報が書かれています。この例では電気モデルの D (Drain)、G (Gate)、S (Source)、DS (Driver Source) ピンの順にピン番号 1、2、3、4 が割り当てられ、さらに熱モデル Tj、Tc、Ta ピンの順にピン番号 Tj、Tc、Ta が割り当てられており、合計 7 ピンの回路シンボルを作成します。7 ピンが揃っていればシンボルのデザインは自由です。作成例を Figure 7 に示します。コンポーネントが 2 つありますが、シミュレーション回路が書きやすいように分離しているため、これで 1 つのシンボルです。




Figure 7. 熱モデルの回路シンボル作成例

熱モデルのネットリストは PSpice 用に記述されているため、SystemVision®Cloud で使用すると構文エラーが発生します。この例ではネットリストの行 55 にある IF 文を次のように VALIF へ修正する必要があります。

原文

```
E20 T0 Ta VALUE={1k*MIN(MAX(IF(TIME>0,(I(V1)-I(V11))*V(1,3)+I(V2)*V(2,3),0),-2MEG),2MEG)}
```

修正後

```
E20 T0 Ta VALUE={1k*MIN(MAX(VALIF(TIME>0,(I(V1)-I(V11))*V(1,3)+I(V2)*V(2,3),0),-2MEG),2MEG)}
```

Figure 8 がシミュレーション回路です。ここではスルーホールパッケージの TO-247-4L (Figure 9) にヒートシンクを装着した状態を想定したシミュレーションを行います。回路図の左半分が電気回路、右半分が熱モデル回路で、SiC MOSFET: SCT3040KR のジャンクション温度 Tj を求めます。

電気回路では SiC MOSFET のゲートに信号源 vsrc2 より矩形波を入力しスイッチング動作しています。電源は vsrc1 より DC 600V が供給され、45Ω の負荷抵抗 r1 を介して SiC MOSFET のドレインへ接続されています。SiC MOSFET がオンするとドレイン電流が約 13A

流れます。このとき SiC MOSFET の導通抵抗とドレイン電流によって導通損失が発生します。SiC MOSFET のゲートとドレインの電圧波形を Figure 10 に示します。

熱モデル回路の Ta ピンには周囲環境温度に対応する電圧源 vsrc3 を接続し電圧 (1℃=1V) を与えます。Tc ピンはパッケージケースでするので、r4 (=ケースとヒートシンク間の熱抵抗 R_{thCF}) と r5 (=ヒートシンクと周囲環境温度間の熱抵抗 R_{thFA}) を介して Ta ピンへ接続します。r4 はケースとヒートシンク間のシリコングリスなどを想定しています。r5 はヒートシンクの熱抵抗です。r4、r5 共に値は抵抗値 (1℃/W=1Ω) で与えます。c1 はヒートシンクの熱容量で Ta ピンへ接続します。r5 と c1 でヒートシンクを構成しています。c1 の熱容量値はヒートシンクのデータシートを参照するか式(2)で概算値を使用します。c1 の値は容量値 (1J/K=1F) で与えます。

$$\text{熱容量 } C = c \times \rho \times V \quad [\text{J/K}] \quad (2)$$

c : 比熱 [kJ/kg · K]

ρ : 密度 [g/cm³]

V : 体積 [cm³]

例えば、アルミニウム製ヒートシンクで、形状が 3cm×3cm×1cm の直方体の場合は次の計算で求めます。

アルミニウムの比熱 c = 0.896 [kJ/kg · K]

アルミニウムの密度 ρ = 2.71 [g/cm³]

体積 V = 3 × 3 × 1 = 9 [cm³]

$$C = 0.896 \times 2.71 \times 9 = 21.8 \text{ [J/K]} = 21.8 \text{ [F]} \quad (3)$$

ジャンクション温度 (=Tj 点の電圧) を求めるために、SiC MOSFET で発生した消費電力を電流に変換して熱モデルの Tj ピンへ印加する必要がありますが、これは前述のように熱モデル内の計算式(1)を介して行われます。なお Tj ピンとグラウンド間に接続されている抵抗 r3 は、Tj ピンがオープンではシミュレーション実行時にエラーが発生するため、計算に影響がない抵抗値で終端処理する目的です。

最後にトランジェント解析を実行し、Tj 点の電圧をモニタすることでジャンクション温度が判ります (1V=1℃)。




Figure 8. シミュレーション回路



Figure 9. TO-247-4L パッケージ




Figure 10. x1 のゲート (下) とドレイン (上) 電圧波形

次に実施例を紹介します。素子のジャンクション温度が 100℃に達すると回路動作を停止し、温度上昇を抑える回路を組み込むことを考えたときに、回路動作可能な時間を求めるシミュレーションを行います。

条件は、形状が 3cm×3cm×1cm のアルミニウム製ヒートシンクを取り付け、r4 はケースとヒートシンク間のシリコングリスで 0.1W/℃の熱抵抗を想定、r5 はヒートシンクの熱抵抗で 15℃/W を想定しています。

す。c1 はヒートシンクの熱容量で、前述の式(3)で計算した値を設定しています。周囲環境温度は 50℃です。

Figure11 に結果を示します。回路動作開始から約 350 秒後にジャンクション温度が 100℃に達することが判りました。




Figure 11. ヒートシンクを取り付けた状態でジャンクション温度が 100℃に達するまでの時間を過渡解析した結果
熱抵抗 15℃/W、形状が 3cm×3cm×1cm のアルミニウム製ヒートシンクを想定
ROHM 製 SiC MOSFET: SCT3040KR

ご 注 意

- 1) 本資料の記載内容は改良などのため予告なく変更することがあります。
- 2) 本資料に記載されている内容は製品のご紹介資料です。ご使用に際しては、別途最新の仕様書を必ずご請求のうえ、ご確認ください。
- 3) ロームは常に品質・信頼性の向上に取り組んでおりますが、半導体製品は種々の要因で故障・誤作動する可能性があります。
万が一、本製品が故障・誤作動した場合であっても、その影響により人身事故、火災損害等が起こらないようご使用機器でのディレーティング、冗長設計、延焼防止、バックアップ、フェイルセーフ等の安全確保をお願いします。定格を超えたご使用や使用上の注意書が守られていない場合、いかなる責任もロームは負うものではありません。
- 4) 本資料に記載されております応用回路例やその定数などの情報につきましては、本製品の標準的な動作や使い方を説明するものです。
したがって、量産設計をされる場合には、外部諸条件を考慮していただきますようお願いいたします。
- 5) 本資料に記載されております技術情報は、製品の代表的動作および応用回路例などを示したものであり、ロームまたは他社の知的財産権その他のあらゆる権利について明示的にも黙示的にも、その実施または利用を許諾するものではありません。上記技術情報の使用に起因して紛争が発生した場合、ロームはその責任を負うものではありません。
- 6) 本資料に掲載されております製品は、耐放射線設計はなされていません。
- 7) 本製品を下記のような特に高い信頼性が要求される機器等に使用される際には、ロームへ必ずご連絡の上、承諾を得てください。
・輸送機器（車載、船舶、鉄道など）、幹線用通信機器、交通信号機器、防災・防犯装置、安全確保のための装置、医療機器、サーバー、太陽電池、送電システム
- 8) 本製品を極めて高い信頼性を要求される下記のような機器等には、使用しないでください。
・航空宇宙機器、原子力制御機器、海底中継機器
- 9) 本資料の記載に従わないために生じたいかなる事故、損害もロームはその責任を負うものではありません。
- 10) 本資料に記載されております情報は、正確を期すため慎重に作成したのですが、万が一、当該情報の誤り・誤植に起因する損害がお客様に生じた場合においても、ロームはその責任を負うものではありません。
- 11) 本製品のご使用に際しては、RoHS 指令など適用される環境関連法令を遵守の上でご使用ください。
お客様がかかる法令を順守しないことにより生じた損害に関して、ロームは一切の責任を負いません。
本製品の RoHS 適合性などの詳細につきましては、セールス・オフィスまでお問合せください。
- 12) 本製品および本資料に記載の技術を輸出又は国外へ提供する際には、「外国為替及び外国貿易法」、「米国輸出管理規則」など適用される輸出関連法令を遵守し、それらの定めにしたがって必要な手続を行ってください。
- 13) 本資料の一部または全部をロームの許可なく、転載・複写することを堅くお断りします。



ローム製品のご検討ありがとうございます。
より詳しい資料やカタログなどご用意しておりますので、お問合せください。

ROHM Customer Support System

<http://www.rohm.co.jp/contact/>